Using the IntervalTools package to validate Lemma 3.4 from Counting Zeros of L-Functions.

In[1]:=
```
<< IntervalTools`
```

In “Counting Zeros of Dirichlet L-Functions”, we define a function $E(a, d, T)$ arising from using Stirling’s Formula to bound the change in the LogGamma function. It is convenient to code in the following terms. The domain is $a \in \{0, 1\}$, $0 \le d \le \frac{9}{2}$, $T \ge 5/7$. In the code in this notebook, we refer to E in paper as GammaSpread in code.

In the paper, Lemma 3.4 asserts that $E(a, d, T)$ is an increasing function of d, and for $\frac{1}{4} \le d \le \frac{5}{8}$ is bounded by

$$\frac{E(a,d,T)}{\pi} \le \frac{(640+216\,a)-112-39\,a}{1536\,(3\,T+3\,a-1)} + \frac{1}{2^{10}}.$$

Defining GammaSpread

In[2]:=

```
e1[a_, d_, T_] = (4 (4 + 3 π)) / (45 ((17 + 2 a)^2 + 4 T^2)^(3/2)) +
    (8 + 6 π) / (45 ((17 + 2 a - 2 d)^2 + 4 T^2)^(3/2)) + (8 + 6 π) / (45 ((17 + 2 a + 2 d)^2 + 4 T^2)^(3/2));
e2[a_, d_, T_] = -((4 T) / (3 ((17 + 2 a)^2 + 4 T^2))) +
    (2 T) / (3 ((17 + 2 a - 2 d)^2 + 4 T^2)) + (2 T) / (3 ((17 + 2 a + 2 d)^2 + 4 T^2)) +
    (1 / 12 696) ((d^2 (T + 94 + 39 a)) / (T^2 + (17 + 2 a)^2));
e3[a_, d_, T_] = (T / 2) Log[1 + (17 + 2 a)^2 / (4 T^2)] -
    (T / 4) Log[1 + (17 + 2 a - 2 d)^2 / (4 T^2)] - (T / 4) Log[1 + (17 + 2 a + 2 d)^2 / (4 T^2)] +
    (1 / 2) ((d^2 (T + 94 + 39 a)) / (T^2 + (17 + 2 a)^2));
e4[shift_, a_, d_, T_] = 2 ArcTan[(shift + 2 a) / (2 T)] -
   ArcTan[(shift + 2 a - 2 d) / (2 T)] - ArcTan[(shift + 2 a + 2 d) / (2 T)];
e5[a_, d_, T_] = (1 / 4) (15 + 2 a - 2 d) ArcTan[(17 + 2 a - 2 d) / (2 T)] +
    (1 / 4) (15 + 2 a + 2 d) ArcTan[(17 + 2 a + 2 d) / (2 T)] -
    (1 / 2) (15 + 2 a) ArcTan[(17 + 2 a) / (2 T)] -
    (1 / 2 + 1 / 12 696) ((d^2 (T + 94 + 39 a)) / (T^2 + (17 + 2 a)^2));

GammaSpread[a_, d_, T_] := e1[a, d, T] + e2[a, d, T] + e3[a, d, T] +
   e4[1, a, d, T] + e4[5, a, d, T] + e4[9, a, d, T] + e4[13, a , d, T] + e5[a, d, T];
GammaSpread[
 a,
 d,
 T]
```

Out[8]=

$$\frac{4\,(4+3\pi)}{45\,\left((17+2a)^2+4T^2\right)^{3/2}} - \frac{4T}{3\,\left((17+2a)^2+4T^2\right)} + \frac{8+6\pi}{45\,\left((17+2a-2d)^2+4T^2\right)^{3/2}} +$$
$$\frac{2T}{3\,\left((17+2a-2d)^2+4T^2\right)} + \frac{8+6\pi}{45\,\left((17+2a+2d)^2+4T^2\right)^{3/2}} + \frac{2T}{3\,\left((17+2a+2d)^2+4T^2\right)} +$$
$$2\,\mathrm{ArcTan}\left[\frac{1+2a}{2T}\right] + 2\,\mathrm{ArcTan}\left[\frac{5+2a}{2T}\right] + 2\,\mathrm{ArcTan}\left[\frac{9+2a}{2T}\right] + 2\,\mathrm{ArcTan}\left[\frac{13+2a}{2T}\right] -$$
$$\frac{1}{2}\,(15+2a)\,\mathrm{ArcTan}\left[\frac{17+2a}{2T}\right] - \mathrm{ArcTan}\left[\frac{1+2a-2d}{2T}\right] - \mathrm{ArcTan}\left[\frac{5+2a-2d}{2T}\right] -$$
$$\mathrm{ArcTan}\left[\frac{9+2a-2d}{2T}\right] - \mathrm{ArcTan}\left[\frac{13+2a-2d}{2T}\right] + \frac{1}{4}\,(15+2a-2d)\,\mathrm{ArcTan}\left[\frac{17+2a-2d}{2T}\right] -$$
$$\mathrm{ArcTan}\left[\frac{1+2a+2d}{2T}\right] - \mathrm{ArcTan}\left[\frac{5+2a+2d}{2T}\right] - \mathrm{ArcTan}\left[\frac{9+2a+2d}{2T}\right] -$$
$$\mathrm{ArcTan}\left[\frac{13+2a+2d}{2T}\right] + \frac{1}{4}\,(15+2a+2d)\,\mathrm{ArcTan}\left[\frac{17+2a+2d}{2T}\right] +$$
$$\frac{1}{2}\,T\,\mathrm{Log}\left[1+\frac{(17+2a)^2}{4T^2}\right] - \frac{1}{4}\,T\,\mathrm{Log}\left[1+\frac{(17+2a-2d)^2}{4T^2}\right] - \frac{1}{4}\,T\,\mathrm{Log}\left[1+\frac{(17+2a+2d)^2}{4T^2}\right]$$

Note that e2, e3, and e5 have terms that cancel. They are included to make nicer interval enclosures.

We need to prove that $E(a, d, T)$ is a monotone increasing function of d, is positive for the entire domain, and is monotone decreasing in T.

GammaSpread[*a*, *d*, *T*] is increasing in *d*

That is, the derivative with respect to *d* is positive. This takes quite a bit of work.

In[9]:=
```
D[GammaSpread[a, d, T], d]
```

Out[9]=

$$\frac{1}{\left(1+\frac{(1+2a-2d)^2}{4T^2}\right)T}+\frac{1}{\left(1+\frac{(5+2a-2d)^2}{4T^2}\right)T}+\frac{1}{\left(1+\frac{(9+2a-2d)^2}{4T^2}\right)T}+\frac{1}{\left(1+\frac{(13+2a-2d)^2}{4T^2}\right)T}-\frac{15+2a-2d}{4\left(1+\frac{(17+2a-2d)^2}{4T^2}\right)T}+\frac{17+2a-2d}{4\left(1+\frac{(17+2a-2d)^2}{4T^2}\right)T}-\frac{1}{\left(1+\frac{(1+2a+2d)^2}{4T^2}\right)T}-\frac{1}{\left(1+\frac{(5+2a+2d)^2}{4T^2}\right)T}-\frac{1}{\left(1+\frac{(9+2a+2d)^2}{4T^2}\right)T}-\frac{1}{\left(1+\frac{(13+2a+2d)^2}{4T^2}\right)T}+\frac{15+2a+2d}{4\left(1+\frac{(17+2a+2d)^2}{4T^2}\right)T}-\frac{17+2a+2d}{4\left(1+\frac{(17+2a+2d)^2}{4T^2}\right)T}+\frac{2(17+2a-2d)(8+6\pi)}{15\left((17+2a-2d)^2+4T^2\right)^{5/2}}+\frac{8(17+2a-2d)T}{3\left((17+2a-2d)^2+4T^2\right)^2}-\frac{2(17+2a+2d)(8+6\pi)}{15\left((17+2a+2d)^2+4T^2\right)^{5/2}}-\frac{8(17+2a+2d)T}{3\left((17+2a+2d)^2+4T^2\right)^2}-\frac{1}{2}\mathrm{ArcTan}\left[\frac{17+2a-2d}{2T}\right]+\frac{1}{2}\mathrm{ArcTan}\left[\frac{17+2a+2d}{2T}\right]$$

We will benefit from introducing a normalizing factor.

In[10]:=
```
Limit[T * D[GammaSpread[a, d, T], d], T → ∞]
```

Out[10]= d

Thus, it is helpful to work with $T * D$[GammaSpread[*a*, *d*, *T*], *d*] instead of the unadorned derivative. We need to demonstrate that the following quantity is nonnegative. We also make the substitution $4\,T^2 \to 1/t$, so that the behavior at $T \to \infty$ is laid bare. Note: $0 \le t \le \frac{1}{4\,(5/7)^2} = \frac{49}{100}$.

In[11]:= `T * D[GammaSpread[a, d, T], d] /. T → 1/(2 Sqrt[t])`

Out[11]=
$$\frac{1}{2\sqrt{t}}\left(\frac{2\,(17+2a-2d)\,(8+6\pi)}{15\left((17+2a-2d)^2+\frac{1}{t}\right)^{5/2}}-\frac{2\,(17+2a+2d)\,(8+6\pi)}{15\left((17+2a+2d)^2+\frac{1}{t}\right)^{5/2}}+\frac{4\,(17+2a-2d)}{3\left((17+2a-2d)^2+\frac{1}{t}\right)^2\sqrt{t}}-\right.$$
$$\frac{4\,(17+2a+2d)}{3\left((17+2a+2d)^2+\frac{1}{t}\right)^2\sqrt{t}}+\frac{2\sqrt{t}}{1+(1+2a-2d)^2\,t}+\frac{2\sqrt{t}}{1+(5+2a-2d)^2\,t}+$$
$$\frac{2\sqrt{t}}{1+(9+2a-2d)^2\,t}+\frac{2\sqrt{t}}{1+(13+2a-2d)^2\,t}-\frac{(15+2a-2d)\sqrt{t}}{2\left(1+(17+2a-2d)^2\,t\right)}+$$
$$\frac{(17+2a-2d)\sqrt{t}}{2\left(1+(17+2a-2d)^2\,t\right)}-\frac{2\sqrt{t}}{1+(1+2a+2d)^2\,t}-\frac{2\sqrt{t}}{1+(5+2a+2d)^2\,t}-$$
$$\frac{2\sqrt{t}}{1+(9+2a+2d)^2\,t}-\frac{2\sqrt{t}}{1+(13+2a+2d)^2\,t}+\frac{(15+2a+2d)\sqrt{t}}{2\left(1+(17+2a+2d)^2\,t\right)}-$$
$$\left.\frac{(17+2a+2d)\sqrt{t}}{2\left(1+(17+2a+2d)^2\,t\right)}-\frac{1}{2}\,\mathrm{ArcTan}\left[(17+2a-2d)\sqrt{t}\right]+\frac{1}{2}\,\mathrm{ArcTan}\left[(17+2a+2d)\sqrt{t}\right]\right)$$

Before proceeding further, we rearrange the expression considerably. Aside from carefully chosen algebra, we use http://fungrim.org/entry/503d4d/

$$\mathrm{ArcTan}[u]-\mathrm{ArcTan}[v]==\mathrm{ArcTan}\left[\frac{u-v}{1+u\,v}\right],\ \text{valid for } u\,v>-1.$$

In our case, $u=(17+2a+2d)\sqrt{t}$ and $v=(17+2a-2d)\sqrt{t}$, so that $u\,v>-1$ for $a\in\{0,1\}$, $t\geq 0$, and $0\leq d\leq 2$. The identity becomes

$$\mathrm{ArcTan}\left[(17+2a+2d)\sqrt{t}\right]-\mathrm{ArcTan}\left[(17+2a-2d)\sqrt{t}\right]==$$
$$\mathrm{ArcTan}\left[\frac{(17+2a+2d)\sqrt{t}-(17+2a-2d)\sqrt{t}}{1+(17+2a+2d)\sqrt{t}\,(17+2a-2d)\sqrt{t}}\right]==\mathrm{ArcTan}\left[\frac{4d}{\frac{1}{\sqrt{t}}+\left((17+2a)^2-4d^2\right)\sqrt{t}}\right]$$

In[12]:=
```
FullSimplify[
  FullSimplify[List @@ Expand[T * D[GammaSpread[a, d, T], d] /. T → 1/(2 Sqrt[t])]] //.
   {x___, b_, c___, y_, e___} /;
     Simplify`SimplifyCount[b /. a → 1] + Simplify`SimplifyCount[y /. a → 1] >
      Simplify`SimplifyCount[Together[b + y /. a → 1]] :→
    {x, c, e, Simplify[Together[b + y]]}]
```

Out[12]=

$$\left\{\frac{1}{1+(1+2a-2d)^2 t}, \frac{1}{1+(5+2a-2d)^2 t}, \frac{1}{1+(9+2a-2d)^2 t}, \frac{1}{1+(13+2a-2d)^2 t}, \frac{1}{2+2(17+2a-2d)^2 t}, -\frac{1}{1+(1+2a+2d)^2 t}, -\frac{1}{1+(5+2a+2d)^2 t}, -\frac{1}{1+(9+2a+2d)^2 t}, -\frac{1}{1+(13+2a+2d)^2 t}, -\frac{1}{2+2(17+2a+2d)^2 t}, \frac{2(17+2a-2d)\,t}{3\left(1+(17+2a-2d)^2 t\right)^2}, -\frac{2(17+2a+2d)\,t}{3\left(1+(17+2a+2d)^2 t\right)^2}, \frac{-\text{ArcTan}\left[(17+2a-2d)\sqrt{t}\right]+\text{ArcTan}\left[(17+2a+2d)\sqrt{t}\right]}{4\sqrt{t}}, \frac{2(17+2a-2d)(4+3\pi)}{15\left((17+2a-2d)^2+\frac{1}{t}\right)^{5/2}\sqrt{t}}, -\frac{2(17+2a+2d)(4+3\pi)}{15\left((17+2a+2d)^2+\frac{1}{t}\right)^{5/2}\sqrt{t}}\right\}$$

In[13]:=

$$\text{edGS1[a\_, d\_, t\_] = Plus @@ }\Big\{\frac{1}{1+(1+2a-2d)^2 t}, \frac{1}{1+(5+2a-2d)^2 t}, \frac{1}{1+(9+2a-2d)^2 t}, \frac{1}{1+(13+2a-2d)^2 t}, -\frac{1}{1+(1+2a+2d)^2 t}, -\frac{1}{1+(5+2a+2d)^2 t}, -\frac{1}{1+(9+2a+2d)^2 t}, -\frac{1}{1+(13+2a+2d)^2 t},$$

$$\text{FullSimplify}\Big[\text{Together}\Big[\frac{2\,(17+2a-2d)\,t}{3\,\big(1+(17+2a-2d)^2 t\big)^2}+\frac{1}{2+2\,(17+2a-2d)^2 t}\Big]\Big],$$

$$\text{FullSimplify}\Big[\text{Together}\Big[-\frac{2\,(17+2a+2d)\,t}{3\,\big(1+(17+2a+2d)^2 t\big)^2}-\frac{1}{2+2\,(17+2a+2d)^2 t}\Big]\Big]\Big\};$$

```
edGS1[a_, d_, 0] = Limit[edGS1[a, d, t], t → 0, Direction → "FromAbove"];
(* happens to be 0, no limit needed*)
```

$$\text{edGS2[a\_, d\_, t\_] = }\frac{1}{4\sqrt{t}}\,\text{ArcTan}\Big[\frac{4\,d}{\frac{1}{\sqrt{t}}+\big((17+2a)^2-4d^2\big)\sqrt{t}}\Big];$$

```
edGS2[a_, d_, 0] = Limit[edGS2[a, d, t], t → 0, Direction → "FromAbove"];
(* happens to be d *)
```

$$\text{edGS3[a\_, d\_, t\_] = }\frac{2\,(4+3\pi)\,t^2}{15}\left(\frac{(17+2a-2d)}{\big((17+2a-2d)^2 t+1\big)^{5/2}}-\frac{(17+2a+2d)}{\big((17+2a+2d)^2 t+1\big)^{5/2}}\right);$$

```
edGS3[a_, d_, 0] = Limit[edGS3[a, d, t], t → 0, Direction → "FromAbove"];
(* happens to be 0, no limit needed*)
dGS[a_, d_, t_] := edGS1[a, d, t] + edGS2[a, d, t] + edGS3[a, d, t];
```

It will be helpful to immediately define a simple interval enclosure for edGS2, as the natural enclosure is badly behaved at $t = 0$.

In[20]:=

$$\text{edGS2[a\_, d\_Interval, t\_Interval] := Interval}\Big[\Big\{\text{Min}\Big[\frac{d\,\pi}{\pi+8\,d\sqrt{t}+\big((17+2a)^2-4d^2\big)\pi\,t}\Big],\ \text{Max}\Big[\frac{d\,\pi^2}{4\,\big(1+2\,d\,\pi\sqrt{t}+(17+2a)^2 t-4d^2 t\big)}\Big]\Big\}\Big];$$

```
(* a lower bound from http://fungrim.org/entry/3fe47b/ *)
(* an upper bound from http://fungrim.org/entry/efebb8/ *)
```

Focus on *a* = 1

For $a = 1$, this function just isn't convex enough to quickly go negative. We consider fixed t, and expand in terms of d. Specifically, we have the following coefficients in a Taylor expansion of dGS[1, d, t] about $d = 0$. The thought is that

$$\text{dGS[1, d, t] = dGS[1, 0, t] + (D[dGS[1, d, t], d] /. d} \to 0) * d + (\text{D[dGS[1, d, t], \{d, 2\}] /. d} \to 0) * \frac{d^2}{2} + (\text{D[dGS[1, d, t], \{d, 3\}] /. d} \to \xi) * \frac{d^3}{3\,!} +$$

for some ξ between 0 and d.

In[21]:=
```
coeffs[1] = With[{a = 1},
  {Limit[dGS[a, d, t], d → 0],
   Simplify[Limit[D[dGS[a, d, t], d], d → 0]], Limit[D[dGS[a, d, t], {d, 2}], d → 0],
   term3step1 = D[dGS[a, d, t], {d, 3}]}]
```

Out[21]=

$$\Big\{0,\ \frac{24\,t}{(1+9\,t)^2}+\frac{56\,t}{(1+49\,t)^2}+\frac{88\,t}{(1+121\,t)^2}+\frac{120\,t}{(1+225\,t)^2}-\frac{236\,t}{3\,(1+361\,t)^2}+\frac{1}{1+361\,t}+\frac{152\,t\,(3+1159\,t)}{3\,(1+361\,t)^3}+\frac{8\,(4+3\,\pi)\,t^2\,(-1+1444\,t)}{15\,(1+361\,t)^{7/2}},\ 0,$$

$$\frac{384\,(3-2\,d)^3\,t^3}{\left(1+(3-2\,d)^2\,t\right)^4}-\frac{192\,(3-2\,d)\,t^2}{\left(1+(3-2\,d)^2\,t\right)^3}+\frac{384\,(7-2\,d)^3\,t^3}{\left(1+(7-2\,d)^2\,t\right)^4}-\frac{192\,(7-2\,d)\,t^2}{\left(1+(7-2\,d)^2\,t\right)^3}+$$

$$\frac{384\,(11-2\,d)^3\,t^3}{\left(1+(11-2\,d)^2\,t\right)^4}-\frac{192\,(11-2\,d)\,t^2}{\left(1+(11-2\,d)^2\,t\right)^3}+\frac{384\,(15-2\,d)^3\,t^3}{\left(1+(15-2\,d)^2\,t\right)^4}-$$

$$\frac{192\,(15-2\,d)\,t^2}{\left(1+(15-2\,d)^2\,t\right)^3}+\frac{96\,(19-2\,d)\,t^2}{\left(1+(19-2\,d)^2\,t\right)^3}+\frac{384\,(3+2\,d)^3\,t^3}{\left(1+(3+2\,d)^2\,t\right)^4}-\frac{192\,(3+2\,d)\,t^2}{\left(1+(3+2\,d)^2\,t\right)^3}+$$

$$\frac{384\,(7+2\,d)^3\,t^3}{\left(1+(7+2\,d)^2\,t\right)^4}-\frac{192\,(7+2\,d)\,t^2}{\left(1+(7+2\,d)^2\,t\right)^3}+\frac{384\,(11+2\,d)^3\,t^3}{\left(1+(11+2\,d)^2\,t\right)^4}-\frac{192\,(11+2\,d)\,t^2}{\left(1+(11+2\,d)^2\,t\right)^3}+$$

$$\frac{384\,(15+2\,d)^3\,t^3}{\left(1+(15+2\,d)^2\,t\right)^4}-\frac{192\,(15+2\,d)\,t^2}{\left(1+(15+2\,d)^2\,t\right)^3}+\frac{96\,(19+2\,d)\,t^2}{\left(1+(19+2\,d)^2\,t\right)^3}+$$

$$\frac{1}{6}\left(3+(61-6\,d)\,(19-2\,d)\,t\right)\left(\frac{1536\,(19-2\,d)^3\,t^3}{\left(1+(19-2\,d)^2\,t\right)^5}-\frac{576\,(19-2\,d)\,t^2}{\left(1+(19-2\,d)^2\,t\right)^4}\right)+$$

$$\frac{1}{2}\left(-2\,(61-6\,d)\,t-6\,(19-2\,d)\,t\right)\left(\frac{96\,(19-2\,d)^2\,t^2}{\left(1+(19-2\,d)^2\,t\right)^4}-\frac{16\,t}{\left(1+(19-2\,d)^2\,t\right)^3}\right)+$$

$$\frac{1}{6}\left(3+(19+2\,d)\,(61+6\,d)\,t\right)\left(\frac{1536\,(19+2\,d)^3\,t^3}{\left(1+(19+2\,d)^2\,t\right)^5}-\frac{576\,(19+2\,d)\,t^2}{\left(1+(19+2\,d)^2\,t\right)^4}\right)+$$

$$\frac{1}{2}\left(6\,(19+2\,d)\,t+2\,(61+6\,d)\,t\right)\left(-\frac{96\,(19+2\,d)^2\,t^2}{\left(1+(19+2\,d)^2\,t\right)^4}+\frac{16\,t}{\left(1+(19+2\,d)^2\,t\right)^3}\right)+\frac{1}{4\,\sqrt{t}}$$

$$\left(\frac{2\left(\frac{32\,d}{\left(\frac{1}{\sqrt{t}}+(361-4\,d^2)\,\sqrt{t}\right)^2}+\frac{256\,d^3\,\sqrt{t}}{\left(\frac{1}{\sqrt{t}}+(361-4\,d^2)\,\sqrt{t}\right)^3}\right)^2\left(\frac{4}{\frac{1}{\sqrt{t}}+(361-4\,d^2)\,\sqrt{t}}+\frac{32\,d^2\,\sqrt{t}}{\left(\frac{1}{\sqrt{t}}+(361-4\,d^2)\,\sqrt{t}\right)^2}\right)}{\left(1+\frac{16\,d^2}{\left(\frac{1}{\sqrt{t}}+(361-4\,d^2)\,\sqrt{t}\right)^2}\right)^3}-\right.$$

$$\Bigg(\Bigg(\frac{4}{\frac{1}{\sqrt{t}}+\left(361-4\,d^2\right)\sqrt{t}}+\frac{32\,d^2\sqrt{t}}{\left(\frac{1}{\sqrt{t}}+\left(361-4\,d^2\right)\sqrt{t}\right)^2}\Bigg)$$

$$\Bigg(\frac{32}{\left(\frac{1}{\sqrt{t}}+\left(361-4\,d^2\right)\sqrt{t}\right)^2}+\frac{1280\,d^2\sqrt{t}}{\left(\frac{1}{\sqrt{t}}+\left(361-4\,d^2\right)\sqrt{t}\right)^3}+\frac{6144\,d^4\,t}{\left(\frac{1}{\sqrt{t}}+\left(361-4\,d^2\right)\sqrt{t}\right)^4}\Bigg)\Bigg)\Big/$$

$$\left(1+\frac{16\,d^2}{\left(\frac{1}{\sqrt{t}}+\left(361-4\,d^2\right)\sqrt{t}\right)^2}\right)^2-$$

$$\frac{2\left(\frac{32\,d}{\left(\frac{1}{\sqrt{t}}+\left(361-4\,d^2\right)\sqrt{t}\right)^2}+\frac{256\,d^3\sqrt{t}}{\left(\frac{1}{\sqrt{t}}+\left(361-4\,d^2\right)\sqrt{t}\right)^3}\right)\left(\frac{96\,d\sqrt{t}}{\left(\frac{1}{\sqrt{t}}+\left(361-4\,d^2\right)\sqrt{t}\right)^2}+\frac{512\,d^3\,t}{\left(\frac{1}{\sqrt{t}}+\left(361-4\,d^2\right)\sqrt{t}\right)^3}\right)}{\left(1+\frac{16\,d^2}{\left(\frac{1}{\sqrt{t}}+\left(361-4\,d^2\right)\sqrt{t}\right)^2}\right)^2}+$$

$$\frac{\frac{96\sqrt{t}}{\left(\frac{1}{\sqrt{t}}+\left(361-4\,d^2\right)\sqrt{t}\right)^2}+\frac{3072\,d^2\,t}{\left(\frac{1}{\sqrt{t}}+\left(361-4\,d^2\right)\sqrt{t}\right)^3}+\frac{12\,288\,d^4\,t^{3/2}}{\left(\frac{1}{\sqrt{t}}+\left(361-4\,d^2\right)\sqrt{t}\right)^4}}{1+\frac{16\,d^2}{\left(\frac{1}{\sqrt{t}}+\left(361-4\,d^2\right)\sqrt{t}\right)^2}}\Bigg)+$$

$$\frac{2}{15}\left(4+3\,\pi\right)t^2\Bigg(\left(19-2\,d\right)\left(\frac{2520\left(19-2\,d\right)^3t^3}{\left(1+\left(19-2\,d\right)^2t\right)^{11/2}}-\frac{840\left(19-2\,d\right)t^2}{\left(1+\left(19-2\,d\right)^2t\right)^{9/2}}\right)-$$

$$6\left(\frac{140\left(19-2\,d\right)^2t^2}{\left(1+\left(19-2\,d\right)^2t\right)^{9/2}}-\frac{20\,t}{\left(1+\left(19-2\,d\right)^2t\right)^{7/2}}\right)-$$

$$\left(19+2\,d\right)\left(-\frac{2520\left(19+2\,d\right)^3t^3}{\left(1+\left(19+2\,d\right)^2t\right)^{11/2}}+\frac{840\left(19+2\,d\right)t^2}{\left(1+\left(19+2\,d\right)^2t\right)^{9/2}}\right)-$$

$$6\left(\frac{140\left(19+2\,d\right)^2t^2}{\left(1+\left(19+2\,d\right)^2t\right)^{9/2}}-\frac{20\,t}{\left(1+\left(19+2\,d\right)^2t\right)^{7/2}}\right)\Bigg)\Bigg\}$$

The first and third coefficients are 0, and, well, that last one is pretty messy.

We now use MooreSkelboe to bound the 2nd coefficient.

In[22]:=
```
{bounds1, where1} = MooreSkelboeMinimize[t ↦ Evaluate[coeffs[1][[2]]],
  t ↦ Evaluate[coeffs[1][[2]]], {Interval[{0, 49/100}]}, 1/2^22]
{N[bounds1, 10], N[where1, 10]}
```

Out[22]=
$$\left\{\text{Interval}\left[\left\{\frac{487623413622108861091945952201614359970806511363513597955616893}{1039251690714019884493118881542137273967719711511515993053593600}+\frac{31833847494041710359375\,(4+3\pi)}{198063482207269222673809408\,\sqrt{17789}},\ \frac{522086401570262421758997016511626561388}{1112700716677185030286277435544849310875}+\frac{904773632\,(4+3\pi)}{5629302740069\,\sqrt{17789}}\right\}\right],\ \left\{\left\{\text{Interval}\left[\left\{\frac{256901}{524288},\frac{49}{100}\right\}\right]\right\},\left\{\text{Interval}\left[\left\{\frac{2055207}{4194304},\frac{256901}{524288}\right\}\right]\right\}\right\}\right\}$$

Out[23]=
```
{Interval[{0.4692224517, 0.4692226713}], {{Interval[{0.4899997710, 0.4900000001}]},
  {Interval[{0.4899995326, 0.4899997712}]}}}
```

We now use MooreSkelboe, after cleaning up the expression some, to bound the 4th coefficient.

In[24]:=
```
term3step2 =
  Map[Numerator[#]/FullSimplify[Denominator[#]] &, List @@ Expand[term3step1]];
```

In[25]:=
```
term3step3 = term3step2 //.
   {x1___, x2_, x3___, x4_, x5___} /; (Denominator[x2] === Denominator[x4]) :>
    {x1, x3, x5, 1/Denominator[x2] FullSimplify[
      Numerator[x2] + Numerator[x4], Assumptions → 0 ≤ t ≤ 49/100 && 0 ≤ d ≤ 2]};
```

In[26]:=
```
term3func[d_, t_] = term3step3;
term3func[d_, 0] := Limit[term3func[d, t], t → 0, Direction → "FromAbove"];
{bounds3, where3} = MooreSkelboeMinimize[term3func, term3func,
  {Interval[{0, 5/9}], Interval[{0, 49/100}]}, 1/128, certificate → False]
```

Out[27]=
$$\left\{\text{Interval}\left[\left\{-\frac{4125086152438841344}{949174187595811623},-\frac{46275898233600}{11031485765581}\right\}\right],\ \left\{\text{Interval}\left[\left\{\frac{491}{896},\frac{11153}{20160}\right\}\right],\text{Interval}\left[\left\{\frac{1973}{4480},\frac{50179}{112000}\right\}\right]\right\}\right\}$$

We now have a lower bound which depends only on *d* and is positive for small *d*.

In[28]:=
```
Reduce[0 ≤ d ≤ 5/9 && 0 + Min[bounds1] * d + 0 * d^2/2 + Min[bounds3] * d^3/3 ≥ 0]
```

Out[28]=
$$0 \le d \le \frac{5}{9}$$

In[29]:=
```
ProveNonNegative[dGS[1, #1, #2] &, dGS[1, #1, #2] &,
  {Interval[{5/9, 9/2}], Interval[{0, 49/100}]}, certificate → False]
```

Out[29]= $\{\{\}, \{\}, \{\{\text{Interval}[\{4, \frac{9}{2}\}], \text{Interval}[\{\frac{1}{4}, \frac{49}{100}\}]\}\}\}$

And now $a = 0$.

In[30]:=
```
coeffs[0] = With[{a = 0},
   {Limit[dGS[a, d, t], d → 0],
    Simplify[Limit[D[dGS[a, d, t], d], d → 0]], Limit[D[dGS[a, d, t], {d, 2}], d → 0],
    term3step1 = D[dGS[a, d, t], {d, 3}]}]
```

Out[30]=
$$\Big\{0, \frac{8t}{(1+t)^2} + \frac{40t}{(1+25t)^2} + \frac{72t}{(1+81t)^2} + \frac{104t}{(1+169t)^2} - \frac{212t}{3(1+289t)^2} + \frac{1}{1+289t} + \frac{136t(3+935t)}{3(1+289t)^3} + \frac{8(4+3\pi)t^2(-1+1156t)}{15(1+289t)^{7/2}}, 0,$$
$$\frac{384(1-2d)^3t^3}{(1+(1-2d)^2t)^4} - \frac{192(1-2d)t^2}{(1+(1-2d)^2t)^3} + \frac{384(5-2d)^3t^3}{(1+(5-2d)^2t)^4} - \frac{192(5-2d)t^2}{(1+(5-2d)^2t)^3} +$$
$$\frac{384(9-2d)^3t^3}{(1+(9-2d)^2t)^4} - \frac{192(9-2d)t^2}{(1+(9-2d)^2t)^3} + \frac{384(13-2d)^3t^3}{(1+(13-2d)^2t)^4} -$$
$$\frac{192(13-2d)t^2}{(1+(13-2d)^2t)^3} + \frac{96(17-2d)t^2}{(1+(17-2d)^2t)^3} + \frac{384(1+2d)^3t^3}{(1+(1+2d)^2t)^4} - \frac{192(1+2d)t^2}{(1+(1+2d)^2t)^3} +$$
$$\frac{384(5+2d)^3t^3}{(1+(5+2d)^2t)^4} - \frac{192(5+2d)t^2}{(1+(5+2d)^2t)^3} + \frac{384(9+2d)^3t^3}{(1+(9+2d)^2t)^4} - \frac{192(9+2d)t^2}{(1+(9+2d)^2t)^3} +$$
$$\frac{384(13+2d)^3t^3}{(1+(13+2d)^2t)^4} - \frac{192(13+2d)t^2}{(1+(13+2d)^2t)^3} + \frac{96(17+2d)t^2}{(1+(17+2d)^2t)^3} +$$
$$\frac{1}{6}(3+(55-6d)(17-2d)t)\left(\frac{1536(17-2d)^3t^3}{(1+(17-2d)^2t)^5} - \frac{576(17-2d)t^2}{(1+(17-2d)^2t)^4}\right) +$$
$$\frac{1}{2}(-2(55-6d)t - 6(17-2d)t)\left(\frac{96(17-2d)^2t^2}{(1+(17-2d)^2t)^4} - \frac{16t}{(1+(17-2d)^2t)^3}\right) +$$
$$\frac{1}{6}(3+(17+2d)(55+6d)t)\left(\frac{1536(17+2d)^3t^3}{(1+(17+2d)^2t)^5} - \frac{576(17+2d)t^2}{(1+(17+2d)^2t)^4}\right) +$$
$$\frac{1}{2}(6(17+2d)t + 2(55+6d)t)\left(-\frac{96(17+2d)^2t^2}{(1+(17+2d)^2t)^4} + \frac{16t}{(1+(17+2d)^2t)^3}\right) + \frac{1}{4\sqrt{t}}$$

$$\left(\frac{2\left(\frac{32\,d}{\left(\frac{1}{\sqrt{t}}+\left(289-4\,d^2\right)\sqrt{t}\right)^2}+\frac{256\,d^3\sqrt{t}}{\left(\frac{1}{\sqrt{t}}+\left(289-4\,d^2\right)\sqrt{t}\right)^3}\right)^2\left(\frac{4}{\frac{1}{\sqrt{t}}+\left(289-4\,d^2\right)\sqrt{t}}+\frac{32\,d^2\sqrt{t}}{\left(\frac{1}{\sqrt{t}}+\left(289-4\,d^2\right)\sqrt{t}\right)^2}\right)}{\left(1+\frac{16\,d^2}{\left(\frac{1}{\sqrt{t}}+\left(289-4\,d^2\right)\sqrt{t}\right)^2}\right)^3}-\right.$$

$$\left(\left(\frac{4}{\frac{1}{\sqrt{t}}+\left(289-4\,d^2\right)\sqrt{t}}+\frac{32\,d^2\sqrt{t}}{\left(\frac{1}{\sqrt{t}}+\left(289-4\,d^2\right)\sqrt{t}\right)^2}\right)\right.$$

$$\left.\left(\frac{32}{\left(\frac{1}{\sqrt{t}}+\left(289-4\,d^2\right)\sqrt{t}\right)^2}+\frac{1280\,d^2\sqrt{t}}{\left(\frac{1}{\sqrt{t}}+\left(289-4\,d^2\right)\sqrt{t}\right)^3}+\frac{6144\,d^4\,t}{\left(\frac{1}{\sqrt{t}}+\left(289-4\,d^2\right)\sqrt{t}\right)^4}\right)\right)\Big/$$

$$\left(1+\frac{16\,d^2}{\left(\frac{1}{\sqrt{t}}+\left(289-4\,d^2\right)\sqrt{t}\right)^2}\right)^2-$$

$$\frac{2\left(\frac{32\,d}{\left(\frac{1}{\sqrt{t}}+\left(289-4\,d^2\right)\sqrt{t}\right)^2}+\frac{256\,d^3\sqrt{t}}{\left(\frac{1}{\sqrt{t}}+\left(289-4\,d^2\right)\sqrt{t}\right)^3}\right)\left(\frac{96\,d\sqrt{t}}{\left(\frac{1}{\sqrt{t}}+\left(289-4\,d^2\right)\sqrt{t}\right)^2}+\frac{512\,d^3\,t}{\left(\frac{1}{\sqrt{t}}+\left(289-4\,d^2\right)\sqrt{t}\right)^3}\right)}{\left(1+\frac{16\,d^2}{\left(\frac{1}{\sqrt{t}}+\left(289-4\,d^2\right)\sqrt{t}\right)^2}\right)^2}+$$

$$\left.\frac{\frac{96\sqrt{t}}{\left(\frac{1}{\sqrt{t}}+\left(289-4\,d^2\right)\sqrt{t}\right)^2}+\frac{3072\,d^2\,t}{\left(\frac{1}{\sqrt{t}}+\left(289-4\,d^2\right)\sqrt{t}\right)^3}+\frac{12\,288\,d^4\,t^{3/2}}{\left(\frac{1}{\sqrt{t}}+\left(289-4\,d^2\right)\sqrt{t}\right)^4}}{1+\frac{16\,d^2}{\left(\frac{1}{\sqrt{t}}+\left(289-4\,d^2\right)\sqrt{t}\right)^2}}\right)+$$

$$\frac{2}{15}\left(4+3\,\pi\right)t^2\left(\left(17-2\,d\right)\left(\frac{2520\left(17-2\,d\right)^3 t^3}{\left(1+\left(17-2\,d\right)^2 t\right)^{11/2}}-\frac{840\left(17-2\,d\right)t^2}{\left(1+\left(17-2\,d\right)^2 t\right)^{9/2}}\right)-\right.$$

$$6\left(\frac{140\left(17-2\,d\right)^2 t^2}{\left(1+\left(17-2\,d\right)^2 t\right)^{9/2}}-\frac{20\,t}{\left(1+\left(17-2\,d\right)^2 t\right)^{7/2}}\right)-$$

$$\left(17+2\,d\right)\left(-\frac{2520\left(17+2\,d\right)^3 t^3}{\left(1+\left(17+2\,d\right)^2 t\right)^{11/2}}+\frac{840\left(17+2\,d\right)t^2}{\left(1+\left(17+2\,d\right)^2 t\right)^{9/2}}\right)-$$

$$\left.6\left(\frac{140\left(17+2\,d\right)^2 t^2}{\left(1+\left(17+2\,d\right)^2 t\right)^{9/2}}-\frac{20\,t}{\left(1+\left(17+2\,d\right)^2 t\right)^{7/2}}\right)\right)\Big\}$$

In[31]:=
```
{bounds1, where1} =
  MooreSkelboeMinimize[t ↦ Evaluate[coeffs[0][[2]]], t ↦ Evaluate[coeffs[0][[2]]],
   {Interval[{0, 49/100}]}, 1/2^20, certificate → False]
```

Out[31]= $\{\text{Interval}[\{\frac{824578130987}{824860999680} + \frac{-4-3\pi}{2061584302080}, 1\}], \{\text{Interval}[\{\frac{941}{2097152}, \frac{943}{2097152}\}]\}\}$

```
In[32]:= term3step2 =
  Map[Numerator[#] / FullSimplify[Denominator[#]] &, List @@ Expand[term3step1]];

In[33]:= term3step3 = term3step2 //.
   {x1___, x2_, x3___, x4_, x5___} /; (Denominator[x2] === Denominator[x4]) :>
    {x1, x3, x5, 1/Denominator[x2] FullSimplify[
      Numerator[x2] + Numerator[x4], Assumptions -> 0 ≤ t ≤ 49/100 && 0 ≤ d ≤ 2]};

In[34]:= term3func[d_, t_] = term3step3;
term3func[d_, 0] := Limit[term3func[d, t], t -> 0, Direction -> "FromAbove"];
{bounds3, where3} = MooreSkelboeMinimize[term3func, term3func,
  {Interval[{0, 2/5}], Interval[{0, 49/100}]}, 1/128, certificate -> False]
```

Out[35]= $\{\text{Interval}[\{-\frac{340613995217471668224}{19396644506332109375}, -\frac{46099200}{3307949}\}],$
$\{\text{Interval}[\{\frac{1513}{9952}, \frac{6363}{39808}\}], \text{Interval}[\{\frac{1135}{2488}, \frac{461481}{995200}\}]\}\}$

```
In[36]:= Reduce[0 ≤ d ≤ 2/5 && 0 + Min[bounds1] * d + 0 * d^2/2 + Min[bounds3] * d^3/3 ≥ 0]
```

Out[36]= $0 \le d \le \frac{2}{5}$

```
In[37]:= ProveNonNegative[dGS[0, #1, #2] &, dGS[0, #1, #2] &,
  {Interval[{2/5, 9/2}], Interval[{0, 49/100}]}, certificate -> False]
```

Out[37]= $\{\{\}, \{\}, \{\{\text{Interval}[\{4, \frac{9}{2}\}], \text{Interval}[\{\frac{1}{4}, \frac{49}{100}\}]\}\}\}$

GammaSpread[*a*, *d*, *T*] is positive

In the paper, we quickly began ignoring the absolute value bars around $E(a, d, T)$. Here, we prove that is legitimate.

```
In[38]:= GammaSpread[a, 0, T] // Together
```

Out[38]= $\frac{8\,(4+3\,\pi)}{45\,(289+68\,a+4\,a^2+4\,T^2)^{3/2}}$

See? Clearly positive for both a and every $T \geq 5/7$. As we know that the derivative (with respect to d) is positive, we have reached the desired conclusion: $E(a, d, T)$ is always positive.

A bound for $\frac{1}{4} \leq d \leq \frac{2}{5}$:

$\frac{1}{\pi}$ GammaSpread[a, d, T] $\leq \frac{(8+a)}{15}\,\frac{d^2}{3T-1+2a}+\frac{1}{128}$

As GammaSpread is monotone in d, we have an easy interval enclosure.

In[39]:=
```
GammaSpread[a_, dint_Interval, T_] :=
  Interval[{Min[GammaSpread[a, Min[dint], T]], Max[GammaSpread[a, Max[dint], T]]}];
```

Now considering T an interval. Actually, we set $T=\frac{1}{2t}$, with $t\in$ Interval$\left[\left\{0,\frac{7}{10}\right\}\right]$. Note that if some function is monotone in t, then it is monotone in T.

e1

In[40]:=
$$\text{Simplify}\left[\text{Expand}\left[\frac{(4+3\pi)\,t^2}{15}\left(\frac{4}{\left((17+2a)^2t^2+1\right)^{5/2}}+\frac{2}{\left((17+2a-2d)^2t^2+1\right)^{5/2}}+\frac{2}{\left((17+2a+2d)^2t^2+1\right)^{5/2}}\right)\right]-\text{D}\left[\text{e1}\left[a,d,\frac{1}{2t}\right],t\right],\text{Assumptions}\to(a==0\,||\,a==1)\,\&\&\,\frac{1}{4}\le d\le\frac{2}{5}\,\&\&\,0\le t\le\frac{7}{10}\right]$$

Out[40]= 0

In[41]:=
$$\frac{(4+3\pi)\,t^2}{15}\left(\frac{4}{\left((17+2a)^2t^2+1\right)^{5/2}}+\frac{2}{\left((17+2a-2d)^2t^2+1\right)^{5/2}}+\frac{2}{\left((17+2a+2d)^2t^2+1\right)^{5/2}}\right)\;/.\;\left\{a\to 0,\,d\to\text{Interval}[\{1/4,\,2/5\}],\,t\to\text{Interval}\left[\left\{0,\frac{7}{10}\right\}\right]\right\}$$

Out[41]= $\text{Interval}\left[\left\{0,\frac{98}{375}(4+3\pi)\right\}\right]$

In[42]:=
$$\frac{(4+3\pi)\,t^2}{15}\left(\frac{4}{\left((17+2a)^2t^2+1\right)^{5/2}}+\frac{2}{\left((17+2a-2d)^2t^2+1\right)^{5/2}}+\frac{2}{\left((17+2a+2d)^2t^2+1\right)^{5/2}}\right)\;/.\;\left\{a\to 1,\,d\to\text{Interval}[\{1/4,\,2/5\}],\,t\to\text{Interval}\left[\left\{0,\frac{7}{10}\right\}\right]\right\}$$

Out[42]= $\text{Interval}\left[\left\{0,\frac{98}{375}(4+3\pi)\right\}\right]$

Thus e1 is monotone in T

In[43]:=
```
e1[a_, d_, Tint_Interval] := Block[{T}, MonotoneEnclosure[e1[a, d, T], T, Tint]];
```

e2

While e2 is not monotone in T, we can use its derivative as a montonicity test.

In[44]:=
$$\text{Simplify}\left[D\left[e2\left[a, d, \frac{1}{2\,t}\right], t\right] == \frac{d^2\left(1+4\,t\left(94+39\,a-(17+2\,a)^2\,t\right)\right)}{6348\left(1+4\,(17+2\,a)^2\,t^2\right)^2} + \frac{1}{3}\left(-\frac{4}{\left(1+(17+2\,a)^2\,t^2\right)^2}+\frac{2}{1+(17+2\,a)^2\,t^2}+\frac{2}{\left(1+(17+2\,a-2\,d)^2\,t^2\right)^2}-\frac{1}{1+(17+2\,a-2\,d)^2\,t^2}+\frac{2}{\left(1+(17+2\,a+2\,d)^2\,t^2\right)^2}-\frac{1}{1+(17+2\,a+2\,d)^2\,t^2}\right),\ \text{Assumptions}\to(a==0\,||\,a==1)\ \&\&\ \frac{1}{4}\le d\le\frac{2}{5}\ \&\&\ 0\le t\le\frac{7}{10}\right]$$

Out[44]= True

In[45]:=
```
e2[a_, d_, Tint_Interval] := Module[{tint = 1/(2 Tint), deriv, crits, vals, ends, t, T},
  deriv[t_] = (d^2 (1 + 4 t (94 + 39 a - (17 + 2 a)^2 t)))/(6348 (1 + 4 (17 + 2 a)^2 t^2)^2) +
    1/3 (-4/(1 + (17 + 2 a)^2 t^2)^2 + 2/(1 + (17 + 2 a)^2 t^2) + 2/(1 + (17 + 2 a - 2 d)^2 t^2)^2 -
      1/(1 + (17 + 2 a - 2 d)^2 t^2) + 2/(1 + (17 + 2 a + 2 d)^2 t^2)^2 - 1/(1 + (17 + 2 a + 2 d)^2 t^2));
  crits = t /. Solve[deriv[t] == 0, t, Reals];
  crits = Select[crits, IntervalMemberQ[tint, #] &];
  ends = {Limit[e2[a, d, T], T → Min[Tint], Direction → "FromAbove"],
    Limit[e2[a, d, T], T → Max[Tint], Direction → "FromBelow"]};
  vals = Union[ends, Map[e2[a, d, 1/(2 #)] &, crits]];
  Interval[{Min[vals], Max[vals]}]];
```

e3

In[46]:=
```
Simplify[e3[a, d, 1/(2 t)] ==
  (d^2 t (1 + 188 t + 78 a t))/(1 + 4 (17 + 2 a)^2 t^2) + 1/(8 t) Log[1 + (8 d^2 t^2 (-1 + ((17 + 2 a)^2 - 2 d^2) t^2))/((1 + (17 + 2 a - 2 d)^2 t^2) (1 + (17 + 2 a + 2 d)^2 t^2))],
  Assumptions → (a == 0 || a == 1) && 1/4 ≤ d ≤ 2/5 && 0 ≤ t ≤ 7/10]
```

Out[46]= True

Not a lot to do, here. Just try plugging in, and also try easy bounds on logarithm!

I use Mathematica’s “Maximize” and “Minimize”,

```
In[47]:= e3[a_, d_, Tint_Interval] :=
  Module[{tint = 1/(2 Tint), t, estimate1, estimate2, outexpr, inexpr, outside, inside},
   outexpr = (d^2 t (1 + 188 t + 78 a t))/(1 + 4 (17 + 2 a)^2 t^2);
   inexpr = (8 d^2 t^2 (-1 + ((17 + 2 a)^2 - 2 d^2) t^2))/((1 + (17 + 2 a - 2 d)^2 t^2) (1 + (17 + 2 a + 2 d)^2 t^2));
   outside = PiecewiseMonotoneEnclosure[
     outexpr, t, t /. Solve[D[outexpr, t] == 0, t, Reals], tint];
   inside = PiecewiseMonotoneEnclosure[inexpr, t,
     t /. Solve[D[inexpr, t] == 0, t, Reals], tint];
   estimate1 = outside + 1/(8 t) Log[1 + inside] /. t → tint;

   (* note that it's just a function of 1 variable at this point *)
   (* use x/(1+x) ≤ Log[1+x] ≤ x *)
   outexpr = (d^2 t (1 + 188 t + 78 a t))/(1 + 4 (17 + 2 a)^2 t^2) + (d^2 t (-1 + ((17 + 2 a)^2 - 2 d^2) t^2))/((1 + (17 + 2 a - 2 d)^2 t^2) (1 + (17 + 2 a + 2 d)^2 t^2));
   outside = PiecewiseMonotoneEnclosure[
     outexpr, t, t /. Solve[D[outexpr, t] == 0, t, Reals], tint];

   inexpr = (d^2 t (1 + 188 t + 78 a t))/(1 + 4 (17 + 2 a)^2 t^2) + (d^2 t (-1 + ((17 + 2 a)^2 - 2 d^2) t^2))/(1 + (17 + 2 a)^2 t^2)^2;
   inside = PiecewiseMonotoneEnclosure[
     inexpr, t, t /. Solve[D[inexpr, t] == 0, t, Reals], tint];

   estimate2 = Interval[{Min[inside], Max[outside]}];

   (*estimate2=Block[{min,max},
      max=First[Maximize[{(d^2 t (1+188 t+78 a t))/(1+4 (17+2 a)^2 t^2)+(d^2 t (-1+((17+2 a)^2-2 d^2) t^2))/((1+(17+2 a-2 d)^2 t^2) (1+(17+2 a+2 d)^2 t^2)),
          Min[tint]≤t≤Max[tint]},t,Reals]];
      min=First[Minimize[{(d^2 t (1+188 t+78 a t))/(1+4 (17+2 a)^2 t^2)+(d^2 t (-1+((17+2 a)^2-2 d^2) t^2))/(1+(17+2 a)^2 t^2)^2,
          Min[tint]≤t≤Max[tint]},t,Reals]];
      Interval[{min,max}]];*)

   IntervalIntersection[estimate1, estimate2]];
```

e4

Fungrim entry 503d4d: ArcTan[x] – ArcTan[y] = ArcTan[$\frac{x-y}{1+x y}$], provided x, y are real and $x y > -1$.

Mathematica is not very good with this identity:

In[48]:=
```
Union[Flatten[Table[Simplify[e4[s, a, d, 1/(2 t)] ==
    ArcTan[(8 d^2 (2 a + s) t^3)/(1 + 2 (2 d^2 + (2 a + s)^2) t^2 + (2 a + s)^2 (2 a - 2 d + s) (2 (a + d) + s) t^4)]],
   {a, 0, 1}, {s, 1, 13, 4}, {d, 1/4, 2/5, 1/100}, {t, 1/100, 7/10, 1/100}]]]
```

Out[48]= {True}

In[49]:=
```
e4[s_, a_, d_, Tint_Interval] :=
  Module[{tint = 1/(2 Tint), inside, crits, vals, ends, t, T},
   inside[t_] = (8 d^2 (2 a + s) t^3)/(1 + 2 (2 d^2 + (2 a + s)^2) t^2 + (2 a + s)^2 (2 a - 2 d + s) (2 (a + d) + s) t^4);
   crits = t /. Solve[D[inside[t], t] == 0, t, Reals];
   crits = Select[crits, IntervalMemberQ[tint, #] &];
   ends = {Limit[e4[s, a, d, T], T → Min[Tint], Direction → "FromAbove"],
     Limit[e4[s, a, d, T], T → Max[Tint], Direction → "FromBelow"]};
   vals = Union[ends, Map[ArcTan[inside[#]] &, crits]];
   Interval[{Min[vals], Max[vals]}]];
```

e5

In[50]:=
```
Simplify[D[e5[a, d, 1/(2 t)], t] == -((15 + 2 a) (17 + 2 a))/(2 + 2 (17 + 2 a)^2 t^2) - (6349 d^2 (1 + 2 (94 + 39 a) t))/(3174 (1 + 4 (17 + 2 a)^2 t^2)^2) +
   (6349 d^2)/(6348 (1 + 4 (17 + 2 a)^2 t^2)) + ((15 + 2 a - 2 d) (17 + 2 a - 2 d))/(4 (1 + (17 + 2 a - 2 d)^2 t^2)) + ((15 + 2 a + 2 d) (17 + 2 a + 2 d))/(4 (1 + (17 + 2 a + 2 d)^2 t^2)),
  Assumptions → (a == 0 || a == 1) && 1/4 ≤ d ≤ 2/5 && 0 ≤ t ≤ 7/10]
```

Out[50]= True

In[51]:=
```
e5[a_, d_, Tint_Interval] := Module[{tint = 1/(2 Tint), deriv, crits, vals, ends, t, T},
   deriv[t_] =
    -((15 + 2 a) (17 + 2 a))/(2 + 2 (17 + 2 a)^2 t^2) - (6349 d^2 (1 + 2 (94 + 39 a) t))/(3174 (1 + 4 (17 + 2 a)^2 t^2)^2) + (6349 d^2)/(6348 (1 + 4 (17 + 2 a)^2 t^2)) +
     ((15 + 2 a - 2 d) (17 + 2 a - 2 d))/(4 (1 + (17 + 2 a - 2 d)^2 t^2)) + ((15 + 2 a + 2 d) (17 + 2 a + 2 d))/(4 (1 + (17 + 2 a + 2 d)^2 t^2));
   crits = t /. Solve[deriv[t] == 0, t, Reals];
   crits = Select[crits, IntervalMemberQ[tint, #] &];
   ends = {Limit[e5[a, d, T], T → Min[Tint], Direction → "FromAbove"],
     Limit[e5[a, d, T], T → Max[Tint], Direction → "FromBelow"]};
   vals = Union[ends, Map[e5[a, d, 1/(2 #)] &, crits]];
   Interval[{Min[vals], Max[vals]}]];
```

The Bound in Lemma 3.4

In[52]:=
$$\frac{1}{3\,T-1+3\,a}\left(-\frac{7}{96}-\frac{13\,a}{512}+\left(\frac{5}{12}+\frac{9\,a}{64}\right)d\right) == \frac{(640+216\,a)\,d-112-39\,a}{1536\,(3\,T+3\,a-1)}\ \text{// Simplify}$$

Out[52]= True

For each value of *a*, verification took about 20 minutes.

In[53]:=
```
AbsoluteTiming[{fa, un, pr} = With[{a = 0}, ProveNonNegative[
      (1/2^10 + 1/(3 * #2 - 1 + 3 a) (-7/96 - (13 a)/512 + (5/12 + (9 a)/64) #1) - GammaSpread[a, #1, #2]/π) &,
      (1/2^10 + 1/(3 * #2 - 1 + 3 a) (-7/96 - (13 a)/512 + (5/12 + (9 a)/64) #1) - GammaSpread[a, #1, #2]/π) &,
      {Interval[{1/4, 5/8}], Interval[{5/7, ∞}]}, MaxDepth → 20]];]
Length /@
 {fa,
  un,
  pr}
```

Out[53]= {1401.76, Null}

Out[54]= {0, 0, 7735}

In[55]:=
```
AbsoluteTiming[{fa, un, pr} = With[{a = 1}, ProveNonNegative[
    (1/2^10 + 1/(3*#2 - 1 + 3 a) (-7/96 - 13 a/512 + (5/12 + 9 a/64) #1) - GammaSpread[a, #1, #2]/π) &,
    (1/2^10 + 1/(3*#2 - 1 + 3 a) (-7/96 - 13 a/512 + (5/12 + 9 a/64) #1) - GammaSpread[a, #1, #2]/π) &,
    {Interval[{1/4, 5/8}], Interval[{5/7, ∞}]}, MaxDepth → 20]];]
Length /@
 {fa,
  un,
  pr}
```

Out[55]= {1123.16, Null}

Out[56]= {0, 0, 5128}

In[57]:=
```
Manipulate[Plot[{1/π GammaSpread[a, d, T], 1/2^10 + 1/(3 T - 1 + 3 a) (-7/96 - 13 a/512 + (5/12 + 9 a/64) d)},
  {d, 1/4, 5/8}, PlotRange → All, ImageSize → 500,
  PlotLegends → {"actual", "bound"}], {a, {0, 1}}, {T, 5/7, 10}]
```

Out[57]=